Rules for integrands of the form $\left(d+e\,x^2\right)^p\,(a+b\,\text{ArcCosh}[c\,x])^n$

1. $\int\left(d+e\,x^2\right)^p\,(a+b\,\text{ArcCosh}[c\,x])^n\,dx$ when $c^2\,d+e==0$

0: $\int(d1+e1\,x)^p\,(d2+e2\,x)^p\,(a+b\,\text{ArcCosh}[c\,x])^n\,dx$ when $d2\,e1+d1\,e2==0\,\land\,p\in\mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $d2\,e1+d1\,e2==0$, then $(d1+e1\,x)\,(d2+e2\,x)==d1\,d2+e1\,e2\,x^2$

Rule: If $d2\,e1+d1\,e2==0\,\land\,p\in\mathbb{Z}$, then

$$\int(d1+e1\,x)^p\,(d2+e2\,x)^p\,(a+b\,\text{ArcCosh}[c\,x])^n\,dx\;\longrightarrow\;\int\left(d1\,d2+e1\,e2\,x^2\right)^p\,(a+b\,\text{ArcCosh}[c\,x])^n\,dx$$

–

Program code:

```
Int[(d1_+e1_.*x_)^p_.*(d2_+e2_.*x_)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  Int[(d1*d2+e1*e2*x^2)^p*(a+b*ArcCosh[c*x])^n,x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,n},x] && EqQ[d2*e1+d1*e2,0] && IntegerQ[p]
```

1. $\int \frac{(a+b\,\text{ArcCosh}[c\,x])^n}{\sqrt{d+e\,x^2}}\,dx$ when $c^2\,d+e==0$

x: $\int \frac{(a+b\,\text{ArcCosh}[c\,x])^n}{\sqrt{d+e\,x^2}}\,dx$ when $c^2\,d+e==0$

- Derivation: Piecewise constant extraction and integration by substitution

- Basis: If $c^2\,d+e==0$, then $\partial_x \frac{\sqrt{1+c\,x}\,\sqrt{-1+c\,x}}{\sqrt{d+e\,x^2}}==0$

- Basis: $\frac{F[\text{ArcCosh}[c\,x]]}{\sqrt{1+c\,x}\,\sqrt{-1+c\,x}}==\frac{1}{c}\,\text{Subst}[F[x],\,x,\,\text{ArcCosh}[c\,x]]\,\partial_x\text{ArcCosh}[c\,x]$

- Note: When $n==1$, this rule would result in a slightly less compact antiderivative since $\int (a+b\,x)^n\,dx$ returns a sum.

- Rule: If $c^2\,d+e==0$, then

$$\int \frac{(a+b\,\text{ArcCosh}[c\,x])^n}{\sqrt{d+e\,x^2}}\,dx \longrightarrow \frac{\sqrt{1+c\,x}\,\sqrt{-1+c\,x}}{c\,\sqrt{d+e\,x^2}}\,\text{Subst}\left[\int (a+b\,x)^n\,dx,\,x,\,\text{ArcCosh}[c\,x]\right]$$

- Program code:

```
(* Int[(a_.+b_.*ArcCosh[c_.*x_])^n_./Sqrt[d_+e_.*x_^2],x_Symbol] :=
  1/c*Simp[Sqrt[1+c*x]*Sqrt[-1+c*x]/Sqrt[d+e*x^2]]*Subst[Int[(a+b*x)^n,x],x,ArcCosh[c*x]] /;
FreeQ[{a,b,c,d,e,n},x] && EqQ[c^2*d+e,0] *)
```

1: $\int \frac{1}{\sqrt{d+e x^2}\,(a+b\,\text{ArcCosh}[c\,x])}\,dx$ when $c^2 d+e==0$

– Derivation: Piecewise constant extraction and integration by substitution

– Basis: If $c^2 d+e==0$, then $\partial_x \frac{\sqrt{1+c x}\,\sqrt{-1+c x}}{\sqrt{d+e x^2}}==0$

– Rule: If $c^2 d+e==0$, then

$$\int \frac{1}{\sqrt{d+e x^2}\,(a+b\,\text{ArcCosh}[c\,x])}\,dx \longrightarrow \frac{\sqrt{1+c x}\,\sqrt{-1+c x}}{b\,c\,\sqrt{d+e x^2}}\,\text{Log}[a+b\,\text{ArcCosh}[c\,x]]$$

– Program code:

```
Int[1/(Sqrt[d_+e_.*x_^2]*(a_.+b_.*ArcCosh[c_.*x_])),x_Symbol] :=
  1/(b*c)*Simp[Sqrt[1+c*x]*Sqrt[-1+c*x]/Sqrt[d+e*x^2]]*Log[a+b*ArcCosh[c*x]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[c^2*d+e,0]
```

```
Int[1/(Sqrt[d1_+e1_.*x_]*Sqrt[d2_+e2_.*x_]*(a_.+b_.*ArcCosh[c_.*x_])),x_Symbol] :=
  1/(b*c)*Simp[Sqrt[1+c*x]/Sqrt[d1+e1*x]]*Simp[Sqrt[-1+c*x]/Sqrt[d2+e2*x]]*Log[a+b*ArcCosh[c*x]] /;
FreeQ[{a,b,c,d1,e1,d2,e2},x] && EqQ[e1,c*d1] && EqQ[e2,-c*d2]
```

2: $\int \frac{(a+b\,\text{ArcCosh}[c\,x])^n}{\sqrt{d+e x^2}}\,dx$ when $c^2 d+e==0 \land n\neq -1$

– Derivation: Piecewise constant extraction and integration by substitution

– Basis: If $c^2 d+e==0$, then $\partial_x \frac{\sqrt{1+c x}\,\sqrt{-1+c x}}{\sqrt{d+e x^2}}==0$

– Rule: If $c^2 d+e==0 \land n\neq -1$, then

$$\int \frac{(a+b\,\text{ArcCosh}[c\,x])^n}{\sqrt{d+e\,x^2}}\,dx \longrightarrow \frac{\sqrt{1+c\,x}\,\sqrt{-1+c\,x}}{b\,c\,(n+1)\,\sqrt{d+e\,x^2}}\,(a+b\,\text{ArcCosh}[c\,x])^{n+1}$$

Program code:

```
Int[(a_.+b_.*ArcCosh[c_.*x_])^n_./Sqrt[d_+e_.*x_^2],x_Symbol] :=
  1/(b*c*(n+1))*Simp[Sqrt[1+c*x]*Sqrt[-1+c*x]/Sqrt[d+e*x^2]]*(a+b*ArcCosh[c*x])^(n+1) /;
FreeQ[{a,b,c,d,e,n},x] && EqQ[c^2*d+e,0] && NeQ[n,-1]
```

```
Int[(a_.+b_.*ArcCosh[c_.*x_])^n_./(Sqrt[d1_+e1_.*x_]*Sqrt[d2_+e2_.*x_]),x_Symbol] :=
  1/(b*c*(n+1))*Simp[Sqrt[1+c*x]/Sqrt[d1+e1*x]]*Simp[Sqrt[-1+c*x]/Sqrt[d2+e2*x]]*(a+b*ArcCosh[c*x])^(n+1) /;
FreeQ[{a,b,c,d1,e1,d2,e2,n},x] && EqQ[e1,c*d1] && EqQ[e2,-c*d2] && NeQ[n,-1]
```

2. $\int (d+e\,x^2)^p\,(a+b\,\text{ArcCosh}[c\,x])^n\,dx$ when $c^2\,d+e==0 \land n>0$

   1: $\int (d+e\,x^2)^p\,(a+b\,\text{ArcCosh}[c\,x])\,dx$ when $c^2\,d+e==0 \land p\in\mathbb{Z}^+$

Derivation: Integration by parts

Basis: $\partial_x\,(a+b\,\text{ArcCosh}[c\,x]) == \frac{b\,c}{\sqrt{1+c\,x}\,\sqrt{-1+c\,x}}$

Rule: If $c^2\,d+e==0 \land p\in\mathbb{Z}^+$, let $u\to\int(d+e\,x^2)^p\,dx$, then

$$\int (d+e\,x^2)^p\,(a+b\,\text{ArcCosh}[c\,x])\,dx \longrightarrow u\,(a+b\,\text{ArcCosh}[c\,x]) - b\,c\int\frac{u}{\sqrt{1+c\,x}\,\sqrt{-1+c\,x}}\,dx$$

Program code:

```
Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCosh[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[(d+e*x^2)^p,x]},
  Dist[a+b*ArcCosh[c*x],u,x] - b*c*Int[SimplifyIntegrand[u/(Sqrt[1+c*x]*Sqrt[-1+c*x]),x],x]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[c^2*d+e,0] && IGtQ[p,0]
```

2. $\int (d+e x^2)^p (a+b \text{ArcCosh}[c x])^n \, dx$ when $c^2 d+e=0 \wedge n>0 \wedge p>0$

1: $\int \sqrt{d+e x^2} \, (a+b \text{ArcCosh}[c x])^n \, dx$ when $c^2 d+e=0 \wedge n>0$

Derivation: Inverted integration by parts

Note: The piecewise constant factor in the second integral reduces the degree of d in the resulting antiderivative.

Rule: If $c^2 d+e=0 \wedge n>0$, then

$$\int \sqrt{d+e x^2} \, (a+b \text{ArcCosh}[c x])^n \, dx \longrightarrow$$

$$\frac{x \sqrt{d+e x^2} \, (a+b \text{ArcCosh}[c x])^n}{2} -$$

$$\frac{b c n \sqrt{d+e x^2}}{2 \sqrt{1+c x} \sqrt{-1+c x}} \int x \, (a+b \text{ArcCosh}[c x])^{n-1} \, dx - \frac{\sqrt{d+e x^2}}{2 \sqrt{1+c x} \sqrt{-1+c x}} \int \frac{(a+b \text{ArcCosh}[c x])^n}{\sqrt{1+c x} \sqrt{-1+c x}} \, dx$$

Program code:

```
Int[Sqrt[d_+e_.*x_^2]*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  x*Sqrt[d+e*x^2]*(a+b*ArcCosh[c*x])^n/2 -
  b*c*n/2*Simp[Sqrt[d+e*x^2]/(Sqrt[1+c*x]*Sqrt[-1+c*x])]*Int[x*(a+b*ArcCosh[c*x])^(n-1),x] -
  1/2*Simp[Sqrt[d+e*x^2]/(Sqrt[1+c*x]*Sqrt[-1+c*x])]*Int[(a+b*ArcCosh[c*x])^n/(Sqrt[1+c*x]*Sqrt[-1+c*x]),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[c^2*d+e,0] && GtQ[n,0]
```

```
Int[Sqrt[d1_+e1_.*x_]*Sqrt[d2_+e2_.*x_]*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  x*Sqrt[d1+e1*x]*Sqrt[d2+e2*x]*(a+b*ArcCosh[c*x])^n/2 -
  b*c*n/2*Simp[Sqrt[d1+e1*x]/Sqrt[1+c*x]]*Simp[Sqrt[d2+e2*x]/Sqrt[-1+c*x]]*
    Int[x*(a+b*ArcCosh[c*x])^(n-1),x] -
  1/2*Simp[Sqrt[d1+e1*x]/Sqrt[1+c*x]]*Simp[Sqrt[d2+e2*x]/Sqrt[-1+c*x]]*
    Int[(a+b*ArcCosh[c*x])^n/(Sqrt[1+c*x]*Sqrt[-1+c*x]),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2},x] && EqQ[e1,c*d1] && EqQ[e2,-c*d2] && GtQ[n,0]
```

2: $\int (d+e x^2)^p (a+b \operatorname{ArcCosh}[c x])^n \, dx$ when $c^2 d+e=0 \wedge n>0 \wedge p>0$

Derivation: Inverted integration by parts

Rule: If $c^2 d+e=0 \wedge n>0 \wedge p>0$, then

$$\int (d+e x^2)^p (a+b \operatorname{ArcCosh}[c x])^n \, dx \longrightarrow$$

$$\frac{x (d+e x^2)^p (a+b \operatorname{ArcCosh}[c x])^n}{2 p+1}+\frac{2 d p}{2 p+1}\int (d+e x^2)^{p-1} (a+b \operatorname{ArcCosh}[c x])^n \, dx-\frac{b c n (d+e x^2)^p}{(2 p+1) (1+c x)^p (-1+c x)^p}\int x (1+c x)^{p-\frac{1}{2}} (-1+c x)^{p-\frac{1}{2}} (a+b \operatorname{ArcCosh}[c x])^{n-1} \, dx$$

–

Program code:

```
Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  x*(d+e*x^2)^p*(a+b*ArcCosh[c*x])^n/(2*p+1) +
  2*d*p/(2*p+1)*Int[(d+e*x^2)^(p-1)*(a+b*ArcCosh[c*x])^n,x] -
  b*c*n/(2*p+1)*Simp[(d+e*x^2)^p/((1+c*x)^p*(-1+c*x)^p)]*
    Int[x*(1+c*x)^(p-1/2)*(-1+c*x)^(p-1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[c^2*d+e,0] && GtQ[n,0] && GtQ[p,0]
```

```
Int[(d1_+e1_.*x_)^p_.*(d2_+e2_.*x_)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  x*(d1+e1*x)^p*(d2+e2*x)^p*(a+b*ArcCosh[c*x])^n/(2*p+1) +
  2*d1*d2*p/(2*p+1)*Int[(d1+e1*x)^(p-1)*(d2+e2*x)^(p-1)*(a+b*ArcCosh[c*x])^n,x] -
  b*c*n/(2*p+1)*Simp[(d1+e1*x)^p/(1+c*x)^p]*Simp[(d2+e2*x)^p/(-1+c*x)^p]*
    Int[x*(1+c*x)^(p-1/2)*(-1+c*x)^(p-1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2},x] && EqQ[e1,c*d1] && EqQ[e2,-c*d2] && GtQ[n,0] && GtQ[p,0]
```

3. $\int (d+e x^2)^p (a+b \operatorname{ArcCosh}[c x])^n \, dx$ when $c^2 d+e=0 \wedge n>0 \wedge p<-1$

1: $\int \frac{(a+b\,\text{ArcCosh}[c\,x])^n}{(d+e\,x^2)^{3/2}}\,dx$ when $c^2 d+e==0 \land n>0$

Derivation: Integration by parts and piecewise constant extraction

- Basis: $\frac{1}{(d+e\,x^2)^{3/2}} == \partial_x \frac{x}{d\sqrt{d+e\,x^2}}$

- Basis: $\partial_x (a+b\,\text{ArcCosh}[c\,x])^n == \frac{b\,c\,n\,(a+b\,\text{ArcCosh}[c\,x])^{n-1}}{\sqrt{1+c\,x}\,\sqrt{-1+c\,x}}$

- Basis: If $c^2 d+e==0$, then $\partial_x \frac{\sqrt{1+c\,x}\,\sqrt{-1+c\,x}}{\sqrt{d+e\,x^2}} == 0$

- Rule: If $c^2 d+e==0 \land n>0$, then

$$\int \frac{(a+b\,\text{ArcCosh}[c\,x])^n}{(d+e\,x^2)^{3/2}}\,dx$$

$$\rightarrow \frac{x\,(a+b\,\text{ArcCosh}[c\,x])^n}{d\sqrt{d+e\,x^2}} - \frac{b\,c\,n}{d}\int \frac{x\,(a+b\,\text{ArcCosh}[c\,x])^{n-1}}{\sqrt{1+c\,x}\,\sqrt{-1+c\,x}\,\sqrt{d+e\,x^2}}\,dx$$

$$\rightarrow \frac{x\,(a+b\,\text{ArcCosh}[c\,x])^n}{d\sqrt{d+e\,x^2}} + \frac{b\,c\,n\,\sqrt{1+c\,x}\,\sqrt{-1+c\,x}}{d\sqrt{d+e\,x^2}}\int \frac{x\,(a+b\,\text{ArcCosh}[c\,x])^{n-1}}{1-c^2 x^2}\,dx$$

Program code:

```
Int[(a_.+b_.*ArcCosh[c_.*x_])^n_./(d_+e_.*x_^2)^(3/2),x_Symbol] :=
  x*(a+b*ArcCosh[c*x])^n/(d*Sqrt[d+e*x^2]) +
  b*c*n/d*Simp[Sqrt[1+c*x]*Sqrt[-1+c*x]/Sqrt[d+e*x^2]]*Int[x*(a+b*ArcCosh[c*x])^(n-1)/(1-c^2*x^2),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[c^2*d+e,0] && GtQ[n,0]
```

```
Int[(a_.+b_.*ArcCosh[c_.*x_])^n_./((d1_+e1_.*x_)^(3/2)*(d2_+e2_.*x_)^(3/2)),x_Symbol] :=
  x*(a+b*ArcCosh[c*x])^n/(d1*d2*Sqrt[d1+e1*x]*Sqrt[d2+e2*x]) +
  b*c*n/(d1*d2)*Simp[Sqrt[1+c*x]/Sqrt[d1+e1*x]]*Simp[Sqrt[-1+c*x]/Sqrt[d2+e2*x]]*
    Int[x*(a+b*ArcCosh[c*x])^(n-1)/(1-c^2*x^2),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2},x] && EqQ[e1,c*d1] && EqQ[e2,-c*d2] && GtQ[n,0]
```

2: $\int \left(d+e x^2\right)^p (a+b \text{ArcCosh}[c x])^n \, dx$ when $c^2 d+e=0 \land n>0 \land p<-1 \land p\neq -\frac{3}{2}$

Rule: If $c^2 d+e=0 \land n>0 \land p<-1 \land p\neq -\frac{3}{2}$, then

$$\int \left(d+e x^2\right)^p (a+b \text{ArcCosh}[c x])^n \, dx \longrightarrow$$

$$-\frac{x \left(d+e x^2\right)^{p+1} (a+b \text{ArcCosh}[c x])^n}{2 d (p+1)}+\frac{2 p+3}{2 d (p+1)} \int \left(d+e x^2\right)^{p+1} (a+b \text{ArcCosh}[c x])^n \, dx-\frac{b c n \left(d+e x^2\right)^p}{2 (p+1) (1+c x)^p (-1+c x)^p} \int x (1+c x)^{p+\frac{1}{2}} (-1+c x)^{p+\frac{1}{2}} (a+b \text{ArcCosh}[c x])^{n-1} \, dx$$

–

Program code:

```
Int[(d_+e_.*x_^2)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  -x*(d+e*x^2)^(p+1)*(a+b*ArcCosh[c*x])^n/(2*d*(p+1)) +
  (2*p+3)/(2*d*(p+1))*Int[(d+e*x^2)^(p+1)*(a+b*ArcCosh[c*x])^n,x] -
  b*c*n/(2*(p+1))*Simp[(d+e*x^2)^p/((1+c*x)^p*(-1+c*x)^p)]*
    Int[x*(1+c*x)^(p+1/2)*(-1+c*x)^(p+1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[c^2*d+e,0] && GtQ[n,0] && LtQ[p,-1] && NeQ[p,-3/2]
```

```
Int[(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  -x*(d1+e1*x)^(p+1)*(d2+e2*x)^(p+1)*(a+b*ArcCosh[c*x])^n/(2*d1*d2*(p+1)) +
  (2*p+3)/(2*d1*d2*(p+1))*Int[(d1+e1*x)^(p+1)*(d2+e2*x)^(p+1)*(a+b*ArcCosh[c*x])^n,x] -
  b*c*n/(2*(p+1))*Simp[(d1+e1*x)^p/(1+c*x)^p]*Simp[(d2+e2*x)^p/(-1+c*x)^p]*
    Int[x*(1+c*x)^(p+1/2)*(-1+c*x)^(p+1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2},x] && EqQ[e1,c*d1] && EqQ[e2,-c*d2] && GtQ[n,0] && LtQ[p,-1] && NeQ[p,-3/2]
```

4: $\int \frac{(a + b\,\text{ArcCosh}[c\,x])^n}{d + e\,x^2}\,dx$ when $c^2 d + e == 0 \land n \in \mathbb{Z}^+$

– Derivation: Integration by substitution

Basis: If $c^2 d + e == 0$, then $\frac{1}{d+e\,x^2} == \frac{1}{c\,d}\,\text{Subst}[\text{Sech}[x],\ x,\ \text{ArcCosh}[c\,x]]\ \partial_x \text{ArcCosh}[c\,x]$

– Note: If $n \in \mathbb{Z}^+$, then $(a + b\,x)^n\,\text{Sech}[x]$ is integrable in closed-form.

– Rule: If $c^2 d + e == 0 \land n \in \mathbb{Z}^+$, then

$$\int \frac{(a + b\,\text{ArcCosh}[c\,x])^n}{d + e\,x^2}\,dx \longrightarrow \frac{1}{c\,d}\,\text{Subst}\left[\int (a + b\,x)^n\,\text{Sech}[x]\,dx,\ x,\ \text{ArcCosh}[c\,x]\right]$$

– Program code:

```
Int[(a_.+b_.*ArcCosh[c_.*x_])^n_./(d_+e_.*x_^2),x_Symbol] :=
  -1/(c*d)*Subst[Int[(a+b*x)^n*Csch[x],x],x,ArcCosh[c*x]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[c^2*d+e,0] && IGtQ[n,0]
```

3: $\int (d+e x^2)^p (a+b \text{ArcCosh}[c x])^n \, dx$ when $c^2 d+e==0 \land n<-1$

Derivation: Integration by parts and piecewise constant extraction

Basis: $\frac{(a+b \text{ArcCosh}[c x])^n}{\sqrt{1+c x} \sqrt{-1+c x}} == \partial_x \frac{(a+b \text{ArcCosh}[c x])^{n+1}}{b c (n+1)}$

Basis: If $c^2 d+e==0$, then $\partial_x \left(\sqrt{1+c x} \sqrt{-1+c x} \left(d+e x^2\right)^p\right) == \frac{c^2 (2 p+1) x \left(d+e x^2\right)^p}{\sqrt{1+c x} \sqrt{-1+c x}}$

Basis: If $c^2 d+e==0$, then $\partial_x \frac{\left(d+e x^2\right)^p}{(1+c x)^p (-1+c x)^p} == 0$

Basis: If $p+\frac{1}{2} \in \mathbb{Z}$, then $(1+c x)^{p-\frac{1}{2}} (-1+c x)^{p-\frac{1}{2}} == \left(-1+c^2 x^2\right)^{p-\frac{1}{2}}$

Rule: If $c^2 d+e==0 \land n<-1$, then

$$\int \left(d+e x^2\right)^p (a+b \text{ArcCosh}[c x])^n \, dx$$

$$\longrightarrow \frac{\sqrt{1+c x} \sqrt{-1+c x} \left(d+e x^2\right)^p (a+b \text{ArcCosh}[c x])^{n+1}}{b c (n+1)} - \frac{c (2 p+1)}{b (n+1)} \int \frac{x \left(d+e x^2\right)^p (a+b \text{ArcCosh}[c x])^{n+1}}{\sqrt{1+c x} \sqrt{-1+c x}} \, dx$$

$$\longrightarrow \frac{\sqrt{1+c x} \sqrt{-1+c x} \left(d+e x^2\right)^p (a+b \text{ArcCosh}[c x])^{n+1}}{b c (n+1)} - \frac{c (2 p+1) \left(d+e x^2\right)^p}{b (n+1) (1+c x)^p (-1+c x)^p} \int x (1+c x)^{p-\frac{1}{2}} (-1+c x)^{p-\frac{1}{2}} (a+b \text{ArcCosh}[c x])^{n+1} \, dx$$

Program code:

```
Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_,x_Symbol] :=
  Simp[Sqrt[1+c*x]*Sqrt[-1+c*x]*(d+e*x^2)^p]*(a+b*ArcCosh[c*x])^(n+1)/(b*c*(n+1)) -
  c*(2*p+1)/(b*(n+1))*Simp[(d+e*x^2)^p/((1+c*x)^p*(-1+c*x)^p)]*
    Int[x*(1+c*x)^(p-1/2)*(-1+c*x)^(p-1/2)*(a+b*ArcCosh[c*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,p},x] && EqQ[c^2*d+e,0] && LtQ[n,-1] && IntegerQ[2*p]
```

```
Int[(d1_+e1_.*x_)^p_.*(d2_+e2_.*x_)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_,x_Symbol] :=
  Sqrt[1+c*x]*Sqrt[-1+c*x]*(d1+e1*x)^p*(d2+e2*x)^p*(a+b*ArcCosh[c*x])^(n+1)/(b*c*(n+1)) -
  c*(2*p+1)/(b*(n+1))*Simp[(d1+e1*x)^p/(1+c*x)^p]*Simp[(d2+e2*x)^p/(-1+c*x)^p]*
    Int[x*(-1+c^2*x^2)^(p-1/2)*(a+b*ArcCosh[c*x])^(n+1),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,p},x] && EqQ[e1,c*d1] && EqQ[e2,-c*d2] && LtQ[n,-1] && IntegerQ[p+1/2]
```

4: $\int (d+e x^2)^p (a+b \text{ArcCosh}[c x])^n \, dx$ when $c^2 d+e==0 \wedge 2 p \in \mathbb{Z}^+$

Derivation: Piecewise constant extraction and integration by substitution

Basis: If $c^2 d+e==0$, then $\partial_x \frac{(d+e x^2)^p}{(1+c x)^p (-1+c x)^p}==0$

Basis: If $2 p \in \mathbb{Z}$, then
$(1+c x)^p (-1+c x)^p == \frac{1}{b c} \text{Subst}\left[\text{Sinh}\left[-\frac{a}{b}+\frac{x}{b}\right]^{2 p+1}, x, a+b \text{ArcCosh}[c x]\right] \partial_x (a+b \text{ArcCosh}[c x])$

Note: If $2 p \in \mathbb{Z}^+$, then $x^n \text{Sinh}\left[-\frac{a}{b}+\frac{x}{b}\right]^{2 p+1}$ is integrable in closed-form.

Rule: If $c^2 d+e==0 \wedge 2 p \in \mathbb{Z}^+$, then

$$\int (d+e x^2)^p (a+b \text{ArcCosh}[c x])^n \, dx$$

$$\longrightarrow \frac{(d+e x^2)^p}{(1+c x)^p (-1+c x)^p} \int (1+c x)^p (-1+c x)^p (a+b \text{ArcCosh}[c x])^n \, dx$$

$$\longrightarrow \frac{(d+e x^2)^p}{b c (1+c x)^p (-1+c x)^p} \text{Subst}\left[\int x^n \text{Sinh}\left[-\frac{a}{b}+\frac{x}{b}\right]^{2 p+1} dx, x, a+b \text{ArcCosh}[c x]\right]$$

Program code:

```
Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  1/(b*c)*Simp[(d+e*x^2)^p/((1+c*x)^p*(-1+c*x)^p)]*Subst[Int[x^n*Sinh[-a/b+x/b]^(2*p+1),x],x,a+b*ArcCosh[c*x]] /;
FreeQ[{a,b,c,d,e,n},x] && EqQ[c^2*d+e,0] && IGtQ[2*p,0]
```

```
Int[(d1_+e1_.*x_)^p_.*(d2_+e2_.*x_)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  1/(b*c)*Simp[(d1+e1*x)^p/(1+c*x)^p]*Simp[(d2+e2*x)^p/(-1+c*x)^p]*Subst[Int[x^n*Sinh[-a/b+x/b]^(2*p+1),x],x,a+b*ArcCosh[c*x]] /;
FreeQ[{a,b,c,d1,e1,d2,e2,n},x] && EqQ[e1,c*d1] && EqQ[e2,-c*d2] && IGtQ[2*p,0]
```

2. $\int (d + e x^2)^p (a + b \text{ArcCosh}[c x])^n \, dx$ when $c^2 d + e \neq 0$

1: $\int (d + e x^2)^p (a + b \text{ArcCosh}[c x]) \, dx$ when $c^2 d + e \neq 0 \land \left(p \in \mathbb{Z}^+ \lor p + \frac{1}{2} \in \mathbb{Z}^-\right)$

Derivation: Integration by parts

Basis: $\partial_x (a + b \text{ArcCosh}[c x]) == \frac{b c}{\sqrt{1+c x} \sqrt{-1+c x}}$

Note: If $p \in \mathbb{Z}^+ \lor p + \frac{1}{2} \in \mathbb{Z}^-$, then $\int (d + e x^2)^p \, dx$ is a rational function.

Rule: If $c^2 d + e \neq 0 \land \left(p \in \mathbb{Z}^+ \lor p + \frac{1}{2} \in \mathbb{Z}^-\right)$, let $u \to \int (d + e x^2)^p \, dx$, then

$$\int (d + e x^2)^p (a + b \text{ArcCosh}[c x]) \, dx \longrightarrow u (a + b \text{ArcCosh}[c x]) - b c \int \frac{u}{\sqrt{1 + c x} \sqrt{-1 + c x}} \, dx$$

Program code:

```
Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCosh[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[(d+e*x^2)^p,x]},
  Dist[a+b*ArcCosh[c*x],u,x] - b*c*Int[SimplifyIntegrand[u/(Sqrt[1+c*x]*Sqrt[-1+c*x]),x],x]] /;
FreeQ[{a,b,c,d,e},x] && NeQ[c^2*d+e,0] && (IGtQ[p,0] || ILtQ[p+1/2,0])
```

x: $\int (d+e x^2)^p (a+b \operatorname{ArcCosh}[c x])^n \, dx$ when $p \in \mathbb{Z}^+$

Derivation: Integration by substitution

Basis: $F[x] == \frac{1}{b c} \operatorname{Subst}\left[F\left[\frac{\operatorname{Cosh}\left[-\frac{a}{b}+\frac{x}{b}\right]}{c}\right] \operatorname{Sinh}\left[-\frac{a}{b}+\frac{x}{b}\right], x, a+b \operatorname{ArcCosh}[c x]\right] \partial_x (a+b \operatorname{ArcCosh}[c x])$

Note: If $p \in \mathbb{Z}^+$, then $x^n \left(c^2 d+e \operatorname{Cosh}\left[-\frac{a}{b}+\frac{x}{b}\right]^2\right)^p \operatorname{Sinh}\left[-\frac{a}{b}+\frac{x}{b}\right]$ is integrable in closed-form.

Rule: If $p \in \mathbb{Z}^+$, then

$$\int (d+e x^2)^p (a+b \operatorname{ArcCosh}[c x])^n \, dx \longrightarrow \frac{1}{b c^{2p+1}} \operatorname{Subst}\left[\int x^n \left(c^2 d+e \operatorname{Cosh}\left[-\frac{a}{b}+\frac{x}{b}\right]^2\right)^p \operatorname{Sinh}\left[-\frac{a}{b}+\frac{x}{b}\right] dx, x, a+b \operatorname{ArcCosh}[c x]\right]$$

Program code:

```
(* Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_,x_Symbol] :=
  1/(b*c^(2*p+1))*Subst[Int[x^n*(c^2*d+e*Cosh[-a/b+x/b]^2)^p*Sinh[-a/b+x/b],x],x,a+b*ArcCosh[c*x]] /;
FreeQ[{a,b,c,d,e,n},x] && IGtQ[p,0] *)
```

3: $\int (d+e x^2)^p (a+b \text{ArcCosh}[c x])^n \, dx$ when $c^2 d+e \neq 0 \land p \in \mathbb{Z} \land (p>0 \lor n \in \mathbb{Z}^+)$

- Derivation: Algebraic expansion

- Rule: If $c^2 d+e \neq 0 \land p \in \mathbb{Z} \land (p>0 \lor n \in \mathbb{Z}^+)$, then

$$\int (d+e x^2)^p (a+b \text{ArcCosh}[c x])^n \, dx \to \int (a+b \text{ArcCosh}[c x])^n \, \text{ExpandIntegrand}\left[(d+e x^2)^p, x\right] dx$$

- Program code:

```
Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*ArcCosh[c*x])^n,(d+e*x^2)^p,x],x] /;
FreeQ[{a,b,c,d,e,n},x] && NeQ[c^2*d+e,0] && IntegerQ[p] && (p>0 || IGtQ[n,0])
```

U: $\int (d+e x^2)^p (a+b \text{ArcCosh}[c x])^n \, dx$

- Rule:

$$\int (d+e x^2)^p (a+b \text{ArcCosh}[c x])^n \, dx \to \int (d+e x^2)^p (a+b \text{ArcCosh}[c x])^n \, dx$$

- Program code:

```
Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  Unintegrable[(d+e*x^2)^p*(a+b*ArcCosh[c*x])^n,x] /;
FreeQ[{a,b,c,d,e,n,p},x]
```

```
Int[(d1_+e1_.*x_)^p_.*(d2_+e2_.*x_)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  Unintegrable[(d1+e1*x)^p*(d2+e2*x)^p*(a+b*ArcCosh[c*x])^n,x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,n,p},x]
```